How can I compare two regression models? I have two OLS regression models (in Stata):
(1) Y = a + b_1 X_1 + b_2 X_2 + fixed effects + e
(2) Y = a + b_1 X_1 + b_2 X_2 + b_3 (X_1 * X_2) + fixed effects + e

In model 1, only b_2 is significant.
In model 2, b_1 and b_3 are weakly significant.
What test can I do to see if model 2 is a "more proper" model than model 1?
Thanks.
 A: Since the OP used linear regression (s)he could better use the F-test rather than the likelihood ratio test. In Stata that means using the test command instead of the lrtest command. In fact, if you only add 1 (interaction) variable, you can just look at the test statistic next to that added variable.
Below I added a simulation that illustrates that the F-test already works in samples as small as 50 observations, where the likelihood ratio test returns $p$-values that don't have the meaning they should have.
clear all
set more off
set seed 123456

program define sim, rclass
    // create data
    drop _all
    set obs 50
    gen x = runiform() < .5
    gen z = rnormal()
    gen y = 1 + .5*x -.25*z + rnormal(0,.25)

    // estimate models
    reg y z x
    est store a
    reg y c.z##i.x
    est store b

    // perform tests
    test 1.x#c.z
    return scalar Fp = r(p)
    lrtest a b
    return scalar lrp = r(p)
end

simulate Fp=r(Fp) lrp=r(lrp) , reps(20000) : sim
simpplot Fp lrp, overall reps(20000)


A: Since the models are nested, i.e. regression (2) is the regression (1) with more variables, you should conduct a Likelihood Ratio test. 
An example in Stata,
reg y x1 x2
est sto model1

reg y x1 x2 x3
est sto model2

lrtest model1 model2

The first model is the null model and the second model is the alternative model.
